


Interface imprimante ZX81

L'interface FLOPPY comporte également une prise pour permettre le raccordement d'une imprimante du type CENTRONICS (à l'aide du CABLE IMPRIMANTE Réf. MAGECO N° 021190).



1 - BRANCHEMENT

Le connecteur du câble imprimante se connecte bord à bord avec le connecteur du floppy et avec le fil rouge de la nappe orienté vers le bord du boîtier interface.

Attention, avant de procéder au branchement mettre hors tension l'alimentation du ZX 81, du FLOPPY et de l'imprimante.

2 - GENERALITES

Les fonctions COPY, LPRINT et LLIST sont utilisables comme pour une imprimante SINCLAIR.

Toutefois grâce à l'interface imprimante vous ne serez plus limité à 32 colonnes, mais en fonction de votre imprimante vous pourrez écrire en 40, 80, 132 colonnes ou plus. Ceci pour les fonctions LPRINT et LLIST.

Par contre la fonction COPY travaille toujours en 32 colonnes car elle reproduit fidèlement le contenu de l'écran.

Les caractères alphabétiques inverses sont remplacés par des minuscules. Quant aux caractères graphiques du ZX 81, vous pourrez :

- soit les conserver, si votre imprimante le permet (GP 80, GP 100, GP 500),
- soit les remplacer par d'autres caractères, dans ce cas les caractères graphiques du ZX 81 sont remplacés par des caractères de ponctuation, et les caractères graphiques inverses sont remplacés par des minuscules accentuées.

3 - PROGRAMMATION DE L'INTERFACE

Les différentes options citées précédemment sont obtenues en exécutant l'instruction POKE 16507,X avant l'exécution d'un LPRINT ou LLIST.

CONFIGURATION DE L'IMPRIMANTE

VALEUR DE X	GR	NGR	>32C	32C	SL	DL
0	OUI		OUI		OUI	
1		OUI	OUI		OUI	
2	OUI			OUI	OUI	
3		OUI		OUI	OUI	
4	OUI		OUI			OUI
5		OUI	OUI			OUI
6	OUI			OUI		OUI
7		OUI		OUI		OUI

GR = Les caractères graphiques du ZX81 seront imprimés.

NGR = Les caractères graphiques du ZX81 seront remplacés par des caractères de ponctuation.

>32C = Le nombre de caractères par ligne pourra dépasser 32 et sera limité par l'imprimante.

32C = Limitation à 32 colonnes.

SL = Caractère en simple largeur.

DL = Caractère en double largeur.

Les explications précédentes se concrétiseront en exécutant le programme suivant :

```
10 FOR M=0 TO 7
20 POKE 16507,M
30 FOR N=0 TO 63
40 LPRINT CHR# N;
50 NEXT N
60 FOR N=128 TO 191
70 LPRINT CHR# N;
80 NEXT N
90 LPRINT
100 LPRINT
110 NEXT M
```

3 - 1 - Saut de ligne

En mode LLIST et LPRINT le saut de ligne se fait toujours avec interligne.

En mode COPY le saut de ligne se fait avec interligne, si le dernier caractère imprimé est un caractère alphanumérique.

Par contre lorsque le dernier caractère est un caractère graphique le saut de ligne se fait sans interligne, ceci permet la reproduction fidèle de ce qui se trouve sur l'écran.

4 - SOUS PROGRAMME D'IMPRESSION

En appelant la routine RAND USR 12000 vous pouvez sortir vers l'imprimante de type CENTRONICS tous les codes ASCII, y compris les codes non imprimables de commande.

Pour cela il faut charger le code à sortir à l'adresse 16507 en utilisant la fonction POKE.

```
Exemple : POKE 16507, 10
          RAND USR 12000
```

Fera exécuter un saut de ligne à l'imprimante.

.../...

5 - IMPRESSION EN HAUTE RESOLUTION

En utilisant la routine situé en 12000 il est possible d'imprimer des courbes en haute résolution lorsque l'imprimante utilisée possède le mode graphique.

L'exemple de programme suivant fonctionne avec la série GP 80, GP 100. Pour les autres imprimantes consulter la notice du fabricant.

```

4 REM DIMENSIONS DE LA ZONE D'IMPRESSION
5 LET XO=200
6 LET YO=60
10 DIM A$(XO,YO/7+1)
15 FAST
20 FOR X=0 TO 199
25 REM FONCTION A REPRESENTER Y=F(X)
30 LET Y=30+30*SIN (X/15)
35 LET Y1=30+30*COS (X/15)
40 LET L=INT (Y/7)
45 LET L1=INT (Y1/7)
50 LET D=INT (Y-7*L)
55 LET D1=INT (Y1-7*L1)
57 REM CONSTITUTION DU TABLEAU
60 LET A$(X+1,L+1)=CHR$(2**D+CODE A$(X+1,L+1))
61 LET A$(X+1,L1+1)=CHR$(2**D1+CODE A$(X+1,L1+1))
65 NEXT X
99 REM DEBUT D'IMPRESSION
100 LPRINT
120 POKE 16507,8
130 RAND USR 12000
140 FOR Y=1 TO YO/7+1
150 FOR X=1 TO XO
170 POKE 16507,128+CODE A$(X,Y)
180 RAND USR 12000
190 NEXT X
200 POKE 16507,10
205 RAND USR 12000
210 NEXT Y

```

Ce programme dessine deux sinusoides.



.../...

Le programme comporte deux phases :

- 1) Création d'un tableau contenant les points de la courbe.
- 2) Envoi des codes du tableau vers l'imprimante.

Explication des variables utilisées

XO = Nombre de points en horizontal du tableau

YO = Nombre de points en vertical du tableau

X = Abscisse

Y = Ordonnée

L = Rang de la ligne

D = Position du point dans la ligne

A\$ = TABLEAU